

## 03 Cuidados com o Posicionamento de Registros

Revisão: 13/07/2002

### Abrangência

<a href="#">Versão 5.07</a>	<a href="#">Versão 5.08</a>	<a href="#">Versão 6.09</a>	<a href="#">Versão 7.10</a>	<a href="#">Versões Anteriores</a>
-----------------------------	-----------------------------	-----------------------------	-----------------------------	------------------------------------

O posicionamento correto de registros é fundamental para a funcionalidade completa dos programas. Algumas dicas para posicionamento de registros são :

Evitar DBGOTOP(), usar DBSEEK(XFILIAL()) para os arquivos de dados do sistema. O comando DBGOTOP() somente será utilizado quando da real necessidade de se efetuar uma leitura desde o início do arquivo independente do tratamento de filial.

Como no mesmo arquivo de dados, poderemos ter registros de várias filiais, desta forma ficará garantido o posicionamento no primeiro registro da filial corrente.

Ao executar um DBSEEK(), verificar se localizou o registro, exemplo:

```
If ! SB1->(dbSeek(xFilial("SB1")))
    // Não achei o registro
Endif
```

Mesmo que seja óbvio a existência do registro, faça o teste pois o programa deve prever que a base de dados não é tão confiável como deveria, e um alerta ajuda a identificar estes casos. Em casos de relatórios, atentar-se para imprimir a mensagem de forma consciente.

Se for executada a função RECLOCK(cAlias, .F.), para alteração do registro atual, em um arquivo no estado de EOF() (caso falhe um DBSEEK() ) será abortado o programa e gravado um arquivo texto de nome MSRLOCK.EOF que poderá ser usado para averiguações.

O comando SOFTSEEK determina se será usada uma busca relativa durante um procura em um banco de dados. Se este comando estiver em ON, e for utilizada uma função DBSEEK(), e nenhuma correspondência for encontrada, o ponteiro de registro ficará no próximo registro do índice que possua um valor mais alto que a expressão utilizada nesta função. Este comando deverá ser utilizado com a máxima atenção, pois caso esteja ligado, poderá localizar um registro errado.

Quanto ao comando DO WHILE não esquecer de incluir a condição referente à filial, quando esta leitura for de registros de uma filial). Exemplo :

```
dbSelectArea("SB1")
dbSeek(xFilial("SB1"))
Do While ! Eof() .And. B1_FILIAL == xFilial("SB1")
    // Processamento
    dbSkip()
Enddo
```

Ao criar uma função que irá desposicionar registros, use a função GETAREA() e RESTAREA(), para voltar tudo à posição original. Exemplo:

```

Dbselectarea("SD1")
aAreasd1 := Getarea() // Armazena o ambiente do arquivo SD1
SD1->(dbsetorder(3))
SD1->(dbseek(xfilial("SD1") + DTOS("01/03/01"), .T.))
Do While ! Eof() .And. D1_FILIAL == xfilial("SD1") .And. DTOS(D1_EMISSAO) <= DTOS(mv_par02)
// Processamento
Dbskip()
Enddo
Restarea(aAreasd1) // Restaura o ambiente do arquivo SD1

```

## Função Posicione

Podemos também buscar uma informação em determinado campo usando apenas uma função.

### Sintaxe:

```
Posicione(cAlias, nOrdem, cChave, cCampo)
```

### Exemplo:

```
Posicione("SB1", 1, xFilial("SB1") + cCodigo, "B1_DESC")
```

Desta forma, será efetuada uma busca no SB1, na ordem 1, chave da busca xFilial("SB1") + cCodigo e será retornado o conteúdo do campo "B1\_DESC". Note que esta função, não restaura a posição original do arquivo alvo (no caso SB1).

É necessário colocar a FILIAL do arquivo na chave passada como parâmetro, caso ela exista na chave do índice.

## Função Existcpo

Retorna se determinada chave existe ou não no arquivo.

### Sintaxe :

```
ExistCpo(cAlias, cChave, nOrdem)
```

### Exemplo :

```
ExistCpo("SB1", 1, cCodigo, "B1_DESC")
```

Desta forma, será efetuada uma busca no SB1, na ordem 1, chave cChave. E será retornado se a chave foi encontrada ou não (.T. ou ,F,). Neste caso não é necessário passar a filial. Ela será inserida automaticamente na chave de pesquisa pela função.

## Restaurando Índice e limpando filtros

Nos relatórios devemos analisar que a função “SetPrint”, possibilita efetuar filtros , escolha da ordem e geração em disco ou impressora , no final dos programas de relatório devemos restaurar a ordem original do arquivos e limpar o filtro e desativar a impressora.

```
//Término do relatorio
dbSelectArea("SRA")
Set Filter to
dbSetOrder(1)
Set Device To Screen
If aReturn[5] = 1
    Set Printer To
    Commit
    ourspool(wnrel)
Endif
MS_FLUSH()
```

### Grupos Relacionados



[Principal / Guias de Referência / Como programar Advpl no ERP](#)

[Topo da Página](#)